

# Descriptive Geometric Kinematic Analysis of Clavel's "Delta" Robot

P.J. Zsombor-Murray  
McGill University  
Department of Mechanical Engineering  
Centre for Intelligent Machines  
Rm. 454, 817 Sherbrooke St. W.  
Montréal (Québec) Canada, H3A 2K6  
<paul@cim.mcgill.ca>

April 1, 2004

**Abstract** Certain high speed industrial assembly robots share a peculiar three legged parallel architecture wherein three "hips", attached to a fixed upper base or "pelvis", actuate "thighs" connected by "knees" to "shins". Each shin is a parallelogramic four-bar linkage. "Ankles" are connected to a common end effector "foot" which executes spatial translatory motion. Inverse and direct kinematic analyses of such manipulators have simple geometric solutions reducible to intersection of line and sphere. Computation is carried out efficiently in a common fixed reference frame.

## 1 Introduction

In 1988 Clavel introduced a three degree of freedom(dof), three identical legged manipulator he called "Delta". This device is shown in Fig. 1. Its end effector(EE) or "foot" executes pure spatial translation.

### 1.1 Description

The fixed frame(FF) or "pelvis" supports three actuated revolute(R) jointed "hips". These R-axes form an equilateral planar triangle. The "knee" end of each "thigh" supports another R-joint whose axis is parallel to the one at the hip. The foot also supports three R-joints whose axes form another triangle which is similar to and maintains the same orientation as the one

on FF. The EE triangle R-axes are held parallel to those on FF because the “shin” is a parallelogram four bar linkage whose R-axes are all perpendicular to the hip, knee and “ankle” R-axes. One pair of linkage R-axes intersects the knee R-axis while the other pair intersects the ankle R-axis.

## 1.2 Kinematic Geometry

It is important to note that when a thigh angle is determined by the actuator, the R-axis of the ankle, if disconnected from EE, would be free to move in the parallel line bundle of the hip R-axis. Note also the three points  $D_i$ ,  $E_i$ ,  $C_i$ ,  $i = 1, 2, 3$ , at hip, knee and ankle of each leg as shown in Fig. IKDELTA.  $D_i$  is the midpoint of a FF R-joint axis triangle side.  $E_i$  is the point on a knee R-joint axis midway between the parallel axis R-joint pair of the four bar linkage while  $C_i$  is midway on the opposite link, coincident with an EE R-joint axis triangle side. If disconnected from the foot,  $C_i$  would move on the sphere centred on  $E_i$ . Similarly, if EE were fixed and  $E_i$  were freed at the knee then  $E_i$  would describe a sphere of the same radius centred on  $C_i$ .

## 1.3 Rationale

Why embark on a reprise of old developments? “Delta” is mentioned in a recent book by Angeles[1]. The elegant symmetry of this robot and its fine, relatively singularity free simplicity, albeit embedded in apparent complexity, were found to be quite compelling. Clavel’s original work, mentioned at the outset, was not cited because it was not readily available for examination however the inverse kinematics of “Delta” was treated by Pierrot[2]. It was implied that a leg chain closure equation approach described therein represented substantial improvement in this regard. His outline of the direct problem relied on three of these quadratic equations. Obviously, numerical solution is required here. Also relevant is work by Hervé[3] wherein a geometrically very similar translational robot, using prismatic rather than revolute actuation, was dealt with. This one, called “Y-Star”, (Notice the transference of the star-delta transformation from its commonly encountered connection with three-phase electrical power.) has screw actuated P-jointed hips. Precise kinematic analysis must be a nightmare because when the plane of a four bar thigh swings on its screw axis, it introduces a parasitic P-joint displacement shift! Only an inverse kinematic analysis was done, this time using the notion of intersecting Schönflies displacement subgroups. In fact, considering the difficult-to-follow analysis and certain obvious alge-

braic errors, it seems the purpose of that article was to expose the liason of groups and not so much to facilitate motion computations. That is the purpose of *this* article: to provide a clear kinematic analysis useful to those who may wish to program and employ nice little three legged robots suited to a line and sphere intersection model. If one is interested in pursuing historical detail and past research on “Delta” and its cousins, many relavent references appear in the three documents listed in the bibliography and cited above.

## 2 Analysis

Now let us examine the inverse and direct kinematics via geometric constructions. These are easy to understand. Computation is based on similar, but not quite identical, geometry.

### 2.1 Inverse Kinematic Construction

Fig. IKDELTA shows a top view of the two triangular platforms. The nine key points,  $D_i, E_i, C_i$ , are clearly visible. The centre of the smaller EE is displaced by  $(x, y, z)$  from origin  $O$  at the centre of FF. Note the design constants.  $e$  is the side length of EE,  $f$  the side of FF,  $r_e$  the distance  $C_i E_i$  and  $r_f$  the distance  $D_i E_i$ . A sphere, radius  $r_e$ , centred on  $C_i$  gives the locus of  $E_i$ . Furthermore a second constraint is imposed by the circular trajectory of  $E_i$  at radius  $r_f$  from centre  $D_i$ . The plane of this circle is visible as a line. The plane cuts the sphere in a small circle in the same plane. In the three auxiliary elevation views one sees this small circle inside the dotted outline of the sphere. The other solid arc is the circle centred on  $D_i$ . The intersection of the arcs yield  $E_i$ , the solution external to FF and EE is chosen as the obviously valid one. The desired actuated R-joint angles are measured from the edge or line view of FF to  $D_i E_i$  as  $\theta_i$ . The leg subscript  $i$  is omitted in all the following equations. It is obvious that a joint angle  $\theta$  must be computed separately for each leg.

### 2.2 Inverse Kinematic Computation

In this case a line will be intersected not with the sphere described above, centred on ankle  $C$ , but with the algebraically simpler one on  $D$ . The homogeneous coordinates  $E\{w : x : y : z\}$  of a point on it are given by

$$(x_{dw} - x)^2 + (y_{dw} - y)^2 + (z_{dw} - z)^2 - r_f w^2 = 0 \quad (1)$$

where  $z_d = 0$  in the frame chosen. Which line? The one through the two desired solutions for  $E$ . It is obtained by intersecting the plane of a thigh circle centred on  $D$  with the plane of a circle produced by the intersection of the sphere given by Eq. 1 and one of radius  $r_e$  centred on  $C$ . The homogeneous coordinates of the three vertical thigh planes,  $\pi\{W_\pi : X_\pi : Y_\pi : Z_\pi\}$  on  $O$  can be written by inspection.

$$\pi_1\{0 : 1 : 0 : 0\}, \quad \pi_2\{0 : 1 : -\sqrt{3} : 0\}, \quad \pi_3\{0 : 1 : \sqrt{3} : 0\}$$

For those not familiar with homogeneous plane coordinates, the last three are normal direction numbers, the first is the moment of the normal direction vector. Since all three planes are on  $O$ , the first coordinates  $W_\pi$  are all zero. Coordinates of the plane of the circle of intersection between the spheres centred on  $C$  and  $D$  are the coefficients of the linear equation which is the difference between the two sphere equations. Its plane coordinates are  $\{W_i : X_i : Y_i : Z_i\}$ . Explicitly, a thigh and shin sphere intersection circle plane has coordinates

$$\{(r_e^2 - r_f^2 + x_d^2 - x_c^2 + y_d^2 - y_c^2 - z_c^2)/2 : (x_c - x_d) : (y_c - y_d) : z_c\}$$

The next step is to compute radial Plücker coordinates of the line, *i.e.*, switching the first and second triplets of the axial coordinates obtained with plane intersection. Expanding on all  $2 \times 2$  minor subdeterminants

$$\left| \begin{array}{cccc} W_\pi & X_\pi & Y_\pi & Z_\pi \\ W & X & Y & Z \end{array} \right| \Rightarrow \{p_{01} : p_{02} : p_{03} : p_{23} : p_{31} : p_{12}\}$$

Finally, for the second constraint, recall the point-on-line relationship.

$$\begin{bmatrix} 0 & p_{23} & p_{31} & p_{12} \\ -p_{23} & 0 & p_{03} & -p_{02} \\ -p_{31} & -p_{03} & 0 & p_{01} \\ -p_{12} & p_{02} & -p_{01} & 0 \end{bmatrix} \begin{bmatrix} w \\ x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (2)$$

The second and third lines of Eq. 2, a doubly rank deficient system of four linear equations, are substituted into Eq. 1 to produce Eq. 3, a quadratic in  $z = z_e$ . This is all that is needed to find  $\theta = \sin^{-1}(z_e/r_f)$ .

$$\begin{aligned}
& \left[ \left( \frac{p_{01}}{p_{03}} \right)^2 + \left( \frac{p_{02}}{p_{03}} \right)^2 \right] z^2 \\
& - 2 \left[ \frac{p_{01}}{p_{03}} \left( \frac{p_{31}}{p_{03}} + x_d \right) - \frac{p_{02}}{p_{03}} \left( \frac{p_{23}}{p_{03}} - y_d \right) \right] wz \\
& + \left[ \left( \frac{p_{31}}{p_{03}} + x_d \right)^2 + \left( \frac{p_{23}}{p_{03}} - y_d \right)^2 + z_d^2 - r_f^2 \right] w^2 = 0 \tag{3}
\end{aligned}$$

Simplifications arise due to choice of frame. Note  $p_{12} = 0$  and for  $z_1$ ,  $p_{01} = p_{31} = 0$  as well. This may be coded at computational expense similar to that incurred in Pierrot's[2] solution however the additional trigonometry of his rotation matrix is not necessary. A programmed example could now be easily presented but that will be saved for the following direct kinematic analysis where the algebraic geometric detail above will not be repeated but the efficacy of a simple algorithm will be shown instead.

### 2.3 Direct Kinematic Construction

Now consider Fig. DKDELTA. EE pose and manipulator design constants are identical to those selected for the inverse kinematic example shown in Fig. IKDELTA. This time the three angles  $\theta_i$  are given instead of the position of the EE centre point, shown as  $O'$ , which must be determined. The geometric key to the direct kinematic solution is the location of points  $E'_i$  which serve as centres of three spheres, radius  $r_e$ . Their two intersection points define two possible solution poses. The constructive solution is shown in a second auxiliary view where the circle of intersection on spheres centered on  $E'_1$  and  $E'_2$  defines a plane which sections the sphere on  $E'_3$  on a second coplanar circle. The intersections of the two circles are obtained here by inspection and projected to the first auxiliary view which shows the FF plane in edge or line view. The lower point is chosen as  $O'$  and the  $z$ -coordinate can be measured here. Projection of this point into the top view provides the other two coordinates. But how are the points  $E'_i$  located? Angles  $\theta_i$  fix  $E_i$  but the free ankles on the shins centred on the fixed knees  $E_i$  describe spheres that contain  $C_i$ , respectively, not  $O'$ . Notice that  $O'$  is located from  $C_i$  by three displacement vectors of constant length and constant direction, pointing inward on EE. Therefore to maintain pure translation of EE,  $O'$  must move on three spheres whose centres are similarly displaced, horizontally inward.

## 2.4 Direct Kinematic Computation

These three displacement vectors  $\mathbf{e}'_i$  are seen to be

$$\mathbf{e}'_1 = \begin{bmatrix} 0 \\ e/(2\sqrt{3}) \\ 0 \end{bmatrix}, \quad \mathbf{e}'_2 = \begin{bmatrix} -e/4 \\ -e/(4\sqrt{3}) \\ 0 \end{bmatrix}, \quad \mathbf{e}'_3 = \begin{bmatrix} e/4 \\ -e/(4\sqrt{3}) \\ 0 \end{bmatrix}$$

and three equations, like Eq. 1, can be written. Differences between the first and second and second and third provide plane coordinates and the key line coordinates to be employed in the computationally simplest of the three sphere equations, like Eq. 3. This is solved for the least  $z$ -coordinate and the second and third lines of Eq. 2 produce the other two coordinates of  $O'$ .

## 2.5 Coded Example

The BASIC program listing below follows the procedure described above except a quadratic univariate in  $x$  was produced by eliminating  $y$  and  $z$  from the sphere equation with the third and fourth lines of Eq. 2.

```

100 INPUT E,F,RE,RF,T1,T2,T3:DTR=3.141592654/180:R3=SQR(3)
110 TR1=DTR*T1:TR2=DTR*T2:TR3=DTR*T3:TR=2*R3
120 EF=(E-F)/TR:E1Y=EF-RF*COS(TR1):E1Z=-RF*SIN(TR1)
130 E2Y=(RF*COS(TR2)-EF)/2:E2X=R3*E2Y:E2Z=-RF*SIN(TR2)
140 E3Y=(RF*COS(TR3)-EF)/2:E3X=-R3*E3Y:E3Z=-RF*SIN(TR3)
150 W1=(EY1^2-E2X^2-EY2^2+E1Z^2-E2Z^2)/2:
    X1=E2X:Y1=E2Y-E1Y:Z1=E2Z-E1Z
160 W2=(E2X^2-E3X^2+E2Y^2-E3Y^2+E2Z^2-E3Z^2)/2:
    X2=E3X-E2X:Y2=E3Y-E2Y:Z2=E3Z-E2Z
170 P01=Y1*Z2-Y2*Z1:P02=Z1*X2-Z2*X1:P03=X1*Y2-X2*Y1:
    IF P01=0 THEN STOP
180 P23=W1*X2-W2*X1:P31=W1*Y2-W2*Y1:P12=W1*Z2-W2*Z1:P2=P01^2
190 A=P2+P02^2+P03^2:T=P12+P01*E1Y:U=P31-P01*E1Z:B=T*P02-U*P03
200 C=T*T+U*U-P2*RE^2:D=B*B-A*C:IF D=0 THEN STOP
210 D=SQR(D):X=(B+D)/A:Y=(P02*X-P12)/P01:Z=(P03*X+P31)/P01
220 PRINT X,Y,Z:STOP:END

```

### 3 Conclusion

It was mentioned at the beginning that this type of manipulator is relatively free of singularity. The the ones that may occur are readily anticipated, like if a leg is fully extended or folded, one experiences three coplanar parallel R-axes. Similarly obvious are the four bar linkage dead-centre singularities. All seem restricted to conditions confined to a leg. However, due to symmetry the conditions may arise simultaneously in all three legs. Carrying out a singularity surface mapping in the kinematic image space may be an interesting exercise but is hardly necessary to achieve fairly trouble free operation of “Delta” type robots. A more interesting issue is the indication that three dof, three legged spatial robots have inverse and direct kinematic solutions of similar complexity. Even six dof robots with three legs, each with two actuators, have, in general, have eight or fewer assembly modes. More to the point, one sees the line and sphere solution paradigm in Stewart-Gough platforms with six P-joint actuated legs where three legs converge to a single S-joint support and two others meet on a rigid body supporting a second S-joint thus creating an effective three legged manipulator. Notwithstanding all this conjecture, it is claimed that the simplest quadratic direct solution for “Delta” manipulators has been exposed for the first time herein.

**Acknowledgement** This work is supported by FCAR (Québec) and NSERC (Canada) research grants.

### References

- [1] Angeles, J. (1997): *Fundamentals of Robotic Mechanical Systems: Theory, Methods & Algorithms*, Springer, ISBN 0-387-94540-7, pp.8-11.
- [2] Pierrot, F., Fournier, A. & Dauchez, P. (1991): “Towards a Fully-Parallel Six DOF Robot for High-Speed Applications”, *Proc. IEEE Rob. & Autom. Conf.*, Sacramento, ISBN 0-8186-2163-X, pp.1288-1293.
- [3] Hervé, J.-M. & Sparacino, F. (1992): “Star, a New Concept in Robotics”, *Proc. 3ARK Wkshp.*, Ferrara, (Parenti-Castelli, V. & Lenarçiç, J. eds.), ISBN 88-86141-00-9, pp.176-183.





FIG. 1

CLAVEL'S "DELTA" ROBOT


+



+ PIC\CL3DIK





+

Three ankle spheres are knee centred  
on inward horizontal offsets  
measured from edge  
to centre of -EE-




+ PIC\CL3DDK

+



+



```

100 INPUT E,F,RE,RF,X0,Y0,Z0:R3=SQR(3):E2=E/2:F2=F/2:E4=E/4:F4=F/4
110 LPRINT E,F,RE,RF:LPRINT X0,Y0,Z0:REM LaTeX11j\DELTAIK 01-01-19
120 C1X=X0:C1Y=Y0-E2/R3:C1Z=Z0:C2X=X0+E4:C2Y=Y0+E4/R3:C2Z=Z0:RF2=RF^2
130 C3X=X0-E4:C3Y=C2Y:C3Z=Z0:D1Y=-F2/R3:D2X=F4:D2Y=F4/R3:D3X=-F4:D3Y=D2Y
EF=RE^2-RF2:X1=C1X:Y1=C1Y-D1Y:Z1=C1Z:W1=(EF-C1X^2+D1Y^2-C1Y^2-C1Z^2)/2
150 X2=C2X-D2X:Y2=C2Y-D2Y:Z2=C2Z:W2=(EF+D2X^2-C2X^2+D2Y^2-C2Y^2-C2Z^2)/2
160 X3=C3X-D3X:Y3=C3Y-D3Y:Z3=C3Z:W3=(EF+D3X^2-C3X^2+D3Y^2-C3Y^2-C3Z^2)/2
170 P02=Z1:P03=-Y1:P23=W1:Q01=-R3*Z2:Q02=-Z2:Q03=Y2+R3*X2:Q23=-W2:Q31=R3*W2
180 R01=R3*Z3:R02=-Z3:R03=Y3-R3*X3:R23=-W3:R31=-R3*W3:RD=180/3.141592654#
190 T=P02/P03:U=P23/P03:A=T*T+1:B=T*(D1Y-U):C=U*(2*D1Y-U)-D1Y^2+RF2
200 D=B*B+A*C:IF D<0 THEN STOP
210 D=SQR(D):V1=(B-D)/A:S=V1/RF:LPRINT RD*ATN(S/SQR(1-S*S))
220 T=Q02/Q03:U=Q01/Q03:V=Q31/Q03:W=Q23/Q03:A=T*T+U*U+1:B=U*(D2X+V)+T*(D2Y-W)
230 C=RF2-D2X*(D2X+2*V)-D2Y*(D2Y-2*W)-V*V-W*W:D=B*B+A*C:IF D<0 THEN STOP
240 D=SQR(D):V2=(B+D)/A:S=V2/RF:LPRINT RD*ATN(S/SQR(1-S*S))
250 T=R02/R03:U=R01/R03:V=R31/R03:W=R23/R03:A=T*T+U*U+1:B=U*(D3X+V)+T*(D3Y-W)
260 C=RF2-D3X*(D3X+2*V)-D3Y*(D3Y-2*W)-V*V-W*W:D=B*B+A*C:IF D<0 THEN STOP
270 D=SQR(D):V3=(B-D)/A:S=V3/RF:LPRINT RD*ATN(S/SQR(1-S*S)):STOP:END
      8          16          10.3094          8
      2          3          -8.7488
-55.19417
-9.49978
-37.79913

```

```

100 INPUT E,F,RE,RF,T1,T2,T3:DTR=3.1415926#/180:R3=SQR(3):REM LaTeX11j\DELTA DK
110 TR1=T1*DTR:TR2=T2*DTR:TR3=T3*DTR:TR=2*R3:REM 01-01-15
120 EF=(E-F)/TR:E1Y=EF-RF*COS(TR1):E1Z=-RF*SIN(TR1)
130 E2Y=(RF*COS(TR2)-EF)/2:E2X=E2Y*R3:E2Z=-RF*SIN(TR2)
) E3Y=(RF*COS(TR3)-EF)/2:E3X=-E3Y*R3:E3Z=-RF*SIN(TR3)
150 W1=(E1Y^2-E2X^2-E2Y^2+E1Z^2-E2Z^2)/2:X1=E2X:Y1=E2Y-E1Y:Z1=E2Z-E1Z
160 W2=(E2X^2-E3X^2+E2Y^2-E3Y^2+E2Z^2-E3Z^2)/2:X2=E3X-E2X:Y2=E3Y-E2Y:Z2=E3Z-E2Z
170 P01=Y1*Z2-Y2*Z1:P02=Z1*X2-Z2*X1:P03=X1*Y2-X2*Y1:IF P01=0 THEN STOP
180 P23=W1*X2-W2*X1:P31=W1*Y2-W2*Y1:P12=W1*Z2-W2*Z1:P2=P01^2
190 A=P2+P02^2+P03^2:B1=P12+P01*E1Y:B2=P31-P01*E1Z:B=P02*B1-P03*B2
200 C=B1^2+B2^2-P2*RE^2:AR=B*B-A*C:IF AR<0 THEN STOP
210 D=SQR(AR):X=(B+D)/A:Y=(P02*X-P12)/P01:Z=(P03*X+P31)/P01
220 LPRINT E,F,RE,RF:LPRINT X,Y,Z:STOP:END
      8          16          10.3094          8
      55.19       9.5         37.8
      2.000059   2.999452   -8.748736

```

+

